

Raport z Laboratorium Układy programowalne 2

Tytuł ćwiczenia: Licznik częstotliwości

Data wykonania: 11 kwietnia 2025

Autor: Patryk Kowalski 267671

Spis treści

1	Cel ćwiczenia	2
2	Przebieg ćwiczenia	2
3	Kod VHDL	2
3.1	Moduł główny - FrequencyCounter	2
3.2	Moduł dzielnika zegara - ClockDivider	4
3.3	Moduł sterownika wyświetlacza - DisplayClockDivider	4
4	Wynik	5
5	Wnioski	6

1 Cel ćwiczenia

Celem ćwiczenia jest zaprojektowanie licznika częstotliwości z akwizycją co 1000 ms.

2 Przebieg ćwiczenia

Do wykonania ćwiczenia wykorzystano do tego płytke Xilinx NEXYS A7, środowisko Vivado, język programowania VHDL. Do przetestowania działania wykorzystano zewnętrzny generator analog discovery 2 wraz z oprogramowaniem Waveforms.

Zaprojektowany licznik składa się z trzech głównych modułów:

- Dzielnik częstotliwości 100 MHz do generowania sygnału 1-sekundowego (ClockDivider)
- Dzielnik częstotliwości do obsługi wyświetlacza (DisplayClockDivider)
- Główny moduł licznika częstotliwości (FrequencyCounter)

Sygnał wejściowy dostarczany jest na pin JB(1) z zewnętrznego generatora. Licznik zlicza impulsy przez 1 sekundę, a następnie wyświetla wartość na 7-segmentowym wyświetlaczu płytki NEXYS A7.

3 Kod VHDL

Poniżej przedstawiono kod VHDL zaprojektowanego licznika.

3.1 Moduł główny - FrequencyCounter

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3 use IEEE.STD_LOGIC_ARITH.ALL;
4 use IEEE.STD_LOGIC_UNSIGNED.ALL;
5
6 entity FrequencyCounter is
7     Port (
8         CLK100MHZ : in STD_LOGIC;
9         JB : in STD_LOGIC_VECTOR(1 to 10);
10        SEG : out STD_LOGIC_VECTOR(6 downto 0);
11        AN : out STD_LOGIC_VECTOR(7 downto 0)
12    );
13 end FrequencyCounter;
14
15 architecture Behavioral of FrequencyCounter is
16     signal one_sec : STD_LOGIC := '0';
17     signal disp_clk : STD_LOGIC := '0';
18     signal pulse_cnt : INTEGER := 0;
19     signal display_value : INTEGER := 0;
20     signal digit_index : INTEGER range 0 to 7 := 0;
21     signal current_digit_value : INTEGER range 0 to 9 := 0;
22     signal JB1_sync1, JB1_sync2 : STD_LOGIC := '0';
23     signal JB1_prev : STD_LOGIC := '0';
24
25     component ClockDivider
26         Port (CLK100MHZ : in STD_LOGIC; ONE_SEC : out STD_LOGIC);
27     end component;
28
29     component DisplayClockDivider
30         Port (CLK100MHZ : in STD_LOGIC; DISP_CLK : out STD_LOGIC);
31     end component;
32
33 begin
34     ClockDiv_Inst : ClockDivider port map (CLK100MHZ => CLK100MHZ, ONE_SEC =>
        one_sec);
```

```

35 DispClk_Inst : DisplayClockDivider port map (CLK100MHZ => CLK100MHZ, DISP_CLK =>
36     disp_clk);
37
38 process (CLK100MHZ)
39 begin
40     if rising_edge(CLK100MHZ) then
41         JB1_sync1 <= JB(1);
42         JB1_sync2 <= JB1_sync1;
43     end if;
44 end process;
45
46 process (CLK100MHZ)
47 begin
48     if rising_edge(CLK100MHZ) then
49         if JB1_prev = '0' and JB1_sync2 = '1' then
50             pulse_cnt <= pulse_cnt + 1;
51         end if;
52         JB1_prev <= JB1_sync2;
53
54         if one_sec = '1' then
55             display_value <= pulse_cnt;
56             pulse_cnt <= 0;
57         end if;
58     end if;
59 end process;
60
61 process (CLK100MHZ)
62 begin
63     if rising_edge(CLK100MHZ) then
64         if disp_clk = '1' then
65             digit_index <= (digit_index + 1) mod 8;
66         end if;
67     end if;
68 end process;
69
70 process (digit_index)
71 begin
72     case digit_index is
73         when 0 => AN <= "11111110";
74         when 1 => AN <= "11111101";
75         when 2 => AN <= "11111011";
76         when 3 => AN <= "11110111";
77         when 4 => AN <= "11101111";
78         when 5 => AN <= "11011111";
79         when 6 => AN <= "10111111";
80         when 7 => AN <= "01111111";
81     end case;
82 end process;
83
84 process (digit_index, display_value)
85 begin
86     case digit_index is
87         when 0 => current_digit_value <= display_value mod 10;
88         when 1 => current_digit_value <= (display_value / 10) mod 10;
89         when 2 => current_digit_value <= (display_value / 100) mod 10;
90         when 3 => current_digit_value <= (display_value / 1000) mod 10;
91         when 4 => current_digit_value <= (display_value / 10000) mod 10;
92         when 5 => current_digit_value <= (display_value / 100000) mod 10;
93         when 6 => current_digit_value <= (display_value / 1000000) mod 10;
94         when 7 => current_digit_value <= (display_value / 10000000) mod 10;
95     end case;
96 end process;

```

```

97     process (current_digit_value)
98     begin
99         case current_digit_value is
100             when 0 => SEG <= "1000000";
101             when 1 => SEG <= "1111001";
102             when 2 => SEG <= "0100100";
103             when 3 => SEG <= "0110000";
104             when 4 => SEG <= "0011001";
105             when 5 => SEG <= "0010010";
106             when 6 => SEG <= "0000010";
107             when 7 => SEG <= "1111000";
108             when 8 => SEG <= "0000000";
109             when 9 => SEG <= "0010000";
110             when others => SEG <= "1111111";
111         end case;
112     end process;
113
114 end Behavioral;

```

Listing 1: Kod modułu głównego licznika częstotliwości

3.2 Moduł dzielnika zegara - ClockDivider

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity ClockDivider is
5      Port (
6          CLK100MHZ : in STD_LOGIC;
7          ONE_SEC : out STD_LOGIC
8      );
9  end ClockDivider;
10
11 architecture Behavioral of ClockDivider is
12     signal clk_div : INTEGER := 0;
13 begin
14     process (CLK100MHZ)
15     begin
16         if rising_edge(CLK100MHZ) then
17             if clk_div = 99999999 then
18                 clk_div <= 0;
19                 ONE_SEC <= '1';
20             else
21                 clk_div <= clk_div + 1;
22                 ONE_SEC <= '0';
23             end if;
24         end if;
25     end process;
26 end Behavioral;

```

Listing 2: Kod modułu dzielnika zegara (1-sekundowy)

3.3 Moduł sterownika wyświetlacza - DisplayClockDivider

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity DisplayClockDivider is
5      Port (
6          CLK100MHZ : in STD_LOGIC;
7          DISP_CLK : out STD_LOGIC

```


```

8   );
9   end DisplayClockDivider;
10
11  architecture Behavioral of DisplayClockDivider is
12      signal clk_div : INTEGER := 0;
13  begin
14      process (CLK100MHZ)
15      begin
16          if rising_edge(CLK100MHZ) then
17              if clk_div = 99999 then
18                  clk_div <= 0;
19                  DISP_CLK <= '1';
20              else
21                  clk_div <= clk_div + 1;
22                  DISP_CLK <= '0';
23              end if;
24          end if;
25      end process;
26  end Behavioral;


```

Listing 3: Kod modułu sterownika wyświetlacza

4 Wynik



Rysunek 1: Zrzut ekranu z programu WaveForms prezentujący sygnał testowy



Rysunek 2: Test z płytką NEXYS A7 i generatorem Analog Discovery 2

5 Wnioski

Zrealizowany układ poprawnie zlicza impulsy zewnętrznego sygnału i wyświetla częstotliwość (liczbę impulsów na sekundę) na wyświetlaczu 7-segmentowym płytki NEXYS A7.